# $Id: PKGBUILD 69678 2012-04-22 12:35:52Z guillaume $
# Maintainer: Guillaume ALAUX <guillaume@archlinux.org>
# Contributor: Jaroslaw Swierczynski <swiergot@aur.archlinux.org>

pkgname=javasqlite
pkgver=20120209
pkgrel=1
pkgdesc="Java support for SQLite database engine"
arch=('i686' 'x86_64')
url="http://www.ch-werner.de/javasqlite/"
license=('custom:BSD-style')
depends=('java-runtime' 'sqlite')
makedepends=('java-environment')
options=('!libtool')
source=(http://www.ch-werner.de/$pkgname/$pkgname-$pkgver.tar.gz)
md5sums=('a5c4f7760ed002749560068b2bd4bb25')
sha1sums=('76aebb0a11b9e1f861ebaeab9ea3e5818b753ab3')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  ./configure --prefix=/usr
  make
}

package() {
  cd "$srcdir/$pkgname-$pkgver"

  make DESTDIR="$pkgdir" install
  install -D -m 644 license.terms "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
